```
> interface(screenwidth=120);
> with(LinearAlgebra):
> P:=x^4+x+1;irreduc(P);
```

$$P := x^4 + x + 1$$

true

```
> A:=1/d*add(a[i]*x^i,i=0..degree(P)-1);
```

$$A := \frac{a_0 + a_1 x + a_2 x^2 + a_3 x^3}{d}$$

```
> H:=d*A;
```

$$H := a_0 + a_1 x + a_2 x^2 + a_3 x^3$$

```
> M:=Matrix([seq([seq(coeff(rem(A*x^i,P,x),x,j),i=0..degree(P)-1)]
  ,j=0..degree(P)-1)]);
```

$$M := \begin{bmatrix} \frac{a_0}{d} & -\frac{a_3}{d} & -\frac{a_2}{d} & -\frac{a_1}{d} \\ \frac{a_1}{d} & \frac{a_0 - a_3}{d} & -\frac{a_3 + a_2}{d} & -\frac{a_2 + a_1}{d} \\ \frac{a_2}{d} & \frac{a_1}{d} & \frac{a_0 - a_3}{d} & -\frac{a_3 + a_2}{d} \\ \frac{a_3}{d} & \frac{a_2}{d} & \frac{a_1}{d} & \frac{a_0 - a_3}{d} \end{bmatrix}$$

```
> cp:=CharacteristicPolynomial(M,x):
> re:=resultant(subs(x=y,P),d*x-subs(x=y,H),y):
> #le poly caract est 1/d^(deg P) * resultant: verification:
> expand(d^(degree(P))*cp-re);
```

$$0$$

```
> P:=x^4+1;
```

$$P := x^4 + 1$$

```
> A:=x^2:H:=A;
```

$$H := x^2$$

```
> M:=Matrix([seq([seq(coeff(rem(A*x^i,P,x),x,j),i=0..degree(P)-1)]
  ,j=0..degree(P)-1)]);
```

$$M := \begin{bmatrix} 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 \\ 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \end{bmatrix}$$

```
> cp:=CharacteristicPolynomial(M,x):
> re:=resultant(subs(x=y,P),x-subs(x=y,H),y):
> #le poly caract est 1/d^(deg P) * resultant: verification:
> expand(cp-re);
```

$$0$$

```
> #le poly min est une puissance de:
> gcd(re,diff(re,x));
```

$$1 + x^2$$

```
> MinimalPolynomial(M,x);#ils sont egaux.
```

$$1 + x^2$$

```
> XT:=t*(t^2-1)^2;YT:=t^2+1;
```

$$XT := t\,(t^2 - 1)^2$$

$$YT := t^2 + 1$$

```
> eq:=resultant(XT-x,YT-y,t);
```

$$eq := x^2 + 16 - 48\,y + 56\,y^2 - 32\,y^3 + 9\,y^4 - y^5$$

```
> with(plots):with(plottools):
Warning, the name changecoords has been redefined

Warning, the name arrow has been redefined

> dessin1:=plot([XT,YT,t=-2..2],color=red):
> #le fait de faire display apres permet d'ajuster l'echelle plus
  facilement
> display(dessin1,view=[-5..5,-1..4]);
```




```
> C1:=x*y-4;C2:=y^2-(x-3)*(x^2-16);
```

$$C1 := y\,x - 4$$

$$C2 := y^2 - (x - 3)(x^2 - 16)$$

```
> d1:=implicitplot(C1,x=-7..7,y=-8..8,color=red,numpoints=2000):
> d2:=implicitplot(C2,x=-6..7,y=-8..8,color=blue,numpoints=2000):
```

```
> display(d1,d2,view=[-6..6,-8..8]);
```



```
> #On elimine y
> eqx:=resultant(C1,C2,y);
```

$$eqx := -48\,x^2 - x^5 + 16\,x^3 + 3\,x^4 + 16$$

```
> factor(eqx,real); # on obtient 5 points reels.
```

$$-1.\,(x+3.981892356)\,(x+0.5365997443)\,(x-0.6633796231)\,(x-2.749235551)\,(x-4.105876927)$$

```
> #c'est de degre 5 car le centre de projection (0,1,0) est
  solution.
> resultant(subs(y=2*x,C1),subs(y=2*x,C2),x);#il n'est pas nul
```

$$6112$$

```
> #NB: A=(1,1) n'est pas sur la droite y=2x. Bt=(t,-2t)
> #eq param de la droite (ABt): x=1+l(t-1),y:=1+l(-2t-1)
> C1t:=subs(x=1+l*(t-1),y=1+l*(-2*t-1),C1);
```

$$C1t := (1 + l\,(-2\,t-1))\,(1 + l\,(t-1)) - 4$$

```
> C2t:=subs(x=1+l*(t-1),y=1+l*(-2*t-1),C2);
```

$$C2t := (1 + l\,(-2\,t-1))^2 - (-2 + l\,(t-1))\,((1 + l\,(t-1))^2 - 16)$$

```
> #la projection est donn{\'e}e par les points de coordonn{\'e}e
  (t,2t) o{\`u} t
> #racine de:
> p2:=resultant(C1t,C2t,l);
```

$$p2 := -14699\,t^6 + 25716\,t^5 + 1779\,t^4 - 15116\,t^3 - 624\,t^2 + 2496\,t + 448$$

```
> # car c'est l'intersection de y=2x avec la droite passant par
  (1,1)
> #et le point a l'infini de Oy qui etait bien dans C1 inter C2
> with(plottools):
> sol:=solve(p2*1.0);
```

$$sol := 1.,\ -0.5062528649,\ -0.3273837129,\ -0.2487685675,\ 0.6001057333,\ 1.231806181$$

```
> pi1:=seq(point([sol[i],-2*sol[i]],color=black),i=1..6):
> di1:=seq((line([1-10*(sol[i]-1),1-10*(-2*sol[i]-1)],[1+10*(sol[i
  ]-1),1+10*(-2*sol[i]-1)],color=green)),i=1..6):
> display(d1,d2,pi1,di1,view=[-7..7,-10..10]);
```



```
> C1:=(x-2)^2+y^2-4;
```

$$C1 := (x-2)^2 + y^2 - 4$$

```
> C2:=y^2-(x-3)*(x^2-16);
```

$$C2 := y^2 - (x-3)\,(x^2-16)$$

```
> d1:=implicitplot(C1,x=-1..5,y=-3..3,color=red,numpoints=1000):
> d2:=implicitplot(C2,x=-5..8,y=-8..8,color=blue,numpoints=1000):
> display(d1,d2,view=[-5..8,-8..8]);
```




```
> #On elimine y
> eqx:=resultant(C1,C2,y);
```

$$eqx := (2\,x^2+20\,x-48-x^3)^2$$

```
> factor(eqx,real); # on obtient 5 points reels.
```

$$(x+4.605551275)^2\,(x-2.605551275)^2\,(x-4.)^2$$

```
> factor(eqx);#On voit qu'il faut introduire le discriminant.
```

$$(x-4)^2\,(x^2+2\,x-12)^2$$

```
> factor(eqx,sqrt(13));
```

$$(x+1+\sqrt{13})^2\,(x+1-\sqrt{13})^2\,(x-4)^2$$

```
> eqred:=gcd(eqx,diff(eqx,x));
```

$$eqred := -2\,x^2-20\,x+48+x^3$$

```
> solx:=[solve(eqred)];
```

$$solx := [4, -1+\sqrt{13}, -1-\sqrt{13}]$$

```
> soly:=seq(gcd(subs(x=solx[i],C1),subs(x=solx[i],C2)),i=1..3);
```

$$soly := y^2,\ 18-6\sqrt{13}+y^2,\ 18+6\sqrt{13}+y^2$$

```
> #les ordonn{\'e}es des points d'abscisse solx[i] sont:
> seq(solve(soly[i]),i=1..3);
```

$$0, 0, \sqrt{-18+6\sqrt{13}}, -\sqrt{-18+6\sqrt{13}}, \sqrt{-18-6\sqrt{13}}, -\sqrt{-18-6\sqrt{13}}$$

```
> #On constate que pour solx[3] les ordonn{\'e}es des points de C1
  inter
> # C2 yant cette abscisse sont complexes conjuguees bien que
  solx[3]
> #soit reel ce qui explique pourquoi le dessin ne nous donnait
  que 4 points.
>
```